

現在価値と内部収益率

SHIMURA Masato

2009年9月28日

目次

1	財務計算の基本公式	1
2	現在価値と内部収益率	4
2.1	割引率（現価係数）	4
2.2	内部収益率 $xirr$	5
2.3	現在価値と将来価値	8
2.4	割引配当モデル DDM	12
2.5	金融不安定仮説と $xirr$	15
3	References	16
付録 A	J のインストールとソースコードの入手	17
A.1	J の入手とインストール	17
A.2	ADDON の入手	17
A.3	スクリプト	17
付録 B	$xirr$	17
付録 C	ロケール	18

概要

プロジェクトの評価法の一つに内部収益率を用いた現在価値や将来価値を求める方法がある。J のアドオンの $irr, xirr$ 関数を用いてプロジェクトの通期の評価法をレビューする。^{*1}

^{*1} 100 円の収入を得るのに必要な費用を計上する営業係数がある。これは単年度の収支を示す係数で、国鉄の末期には盛んに見受けられたが、景気が良くなるとほとんど公表されなくなる。

累積支払額	<i>accndx</i>	<i>accumulate indexed payments.</i>
	<i>accpay</i>	<i>accumulate payments</i>
支払い	<i>amort</i>	<i>calculate amortization table</i>
	<i>osb</i>	<i>outstanding loan balance</i>
年金	<i>annt</i>	<i>annuities certain</i>
正味現在価値	<i>npv</i>	<i>net present value</i>
	<i>vt</i>	<i>v to the powert</i>
内部利益率	<i>irr</i>	<i>internal rate of return</i>
	<i>xirr</i>	<i>as irr but with non – periodic payments</i>
78 法	<i>ruleof78</i>	<i>outstanding balance on rule of 78</i>
Util	<i>accint</i>	<i>accumulate intlist for duration</i>
	<i>effnom</i>	<i>return effective from nominal</i>
	<i>intexpand</i>	<i>expand modal intlist</i>
	<i>intm</i>	<i>convert single intrate to modal</i>
	<i>intrep</i>	<i>expand intlist</i>
	<i>vrep</i>	<i>expand intlist to discount rates</i>
	<i>stretch</i>	<i>stretch list to length len</i>

- 自作の Script を併用する。

(かつて作成したもの) require jpath '~user/classes/finance/square_6.ijs' NB.
adjust any folder

(新作)require jpath '~user/classes/finance/finance0.ijs' NB. adjust any folder

2 現在価値と内部収益率

2.1 割引率 (現価係数)

割引率

$$(1 + r)^{-t}$$

- 金利 5 7 10 15 20 NB. r (%)
- 期間 5 10 15 20 年 NB. t

複利原価表: $FPCV = (1 + r)^{-t}$

利子率毎の割引率を計算する。縦の見出しが金利、横の見出しが期間である。

```
5 7 10 17 20 find_dp_sub table 5 10 15 20
```

find_dp_sub	5	10	15	20
5	0.783526	0.613913	0.481017	0.376889
7	0.712986	0.508349	0.362446	0.258419
10	0.620921	0.385543	0.239392	0.148644
17	0.456111	0.208037	0.0948882	0.0432796
20	0.401878	0.161506	0.0649055	0.0260841

(縄田 p206)

Script

```
find_dp=: 4 : 0
```

```
NB. Usage x. find_dp y
```

```
(_,y ), x,. x find_dp_sub y
```

```
)
```

$$\frac{1}{(1 + \frac{r}{100})^n}$$

```
find_dp_sub=: 4 : ' % (>: x % 100) ^/ y '
```

*3

*3 table appendix 参照

2.2 内部収益率 *xirr*

- J の addon にある *irr,xirr* を用いる。
- *irr* Internal rate of Return 内部収益率
- *xirr* は途中でのキャッシュフローの変更ができる。

内部収益率は投資プロジェクトの評価指標として用いることができる。
投資に対する将来のキャッシュフローの現在価値と、投資額の現在価値とが丁度等しくなる割引率
(内部収益率)

内部収益率が資本コストよりも大であればその投資は有利である。

$$\sum_{t=1}^T \frac{A_t}{(1+r^*)^t} - C = 0$$

この式は $irr(y^*)$ について解析的に解くことは出来ないので数値計算により求める。

2.2.1 内部収益率の関数 *irr,xirr*

- *irr,xirr* はニュートン・ラフソン法による反復解法を用いている。
- input 通算年数とキャッシュフローを用いる。長期投資も計算できる
- 減価償却は将来の再投資のための会計費用であるがキャッシュフローには含めない。^{*4}

```
NB. time ,: cashflow (pay and income)
```

```
]a=: 0 1 2 3 ,: _100 5 5 125
```

```
0 1 2 3 NB. time
```

```
_100 5 5 125 NB. cf
```

- Calc
 $\frac{5}{1+r} + \frac{5}{(1+r)^2} + \frac{5}{(1+r)^3} = 100$ を求める。

```
irr _100 5 5 125
```

```
0.109851
```

```
xirr a
```

```
0.109851
```

- J Grammar

^{*4} 含めた数値計算も可能である。

- *irr, xirr* とタイプするとスクリプトが顕れる

- 層連結 (ラミネート) (, :)

```
1 2 3, :4 5 6
```

```
1 2 3
```

```
4 5 6
```

ラミネートにより 2 段のデータとする。上下のデータの個数は一致しなければならない。

- *irr* は *time* を受け付けない。

```
irr a
```

```
0.140477
```

```
irr 0 1 2 3 _100 5 5 125
```

```
0.140477
```

- *irr, xirr* の左パラメータ (*x*) は反復回数。反復回数を特に指定したいときにのみ用いる (通常は指定しないで単項関数として用いる)

追加した SCRIPT .

DF 割引係数 <i>discount factor</i>	$DF = \frac{1}{(1 + xirr)^t}$ 2.1 参照 <i>xirr</i> を用いる	<pre>df 0 1 2 3, :_100 5 5 125 1 0.901022 0.81184 0.731486</pre>
PV 現在価値 <i>present value</i>	$PV = CF \times DF$ <i>CF</i> Cash flow このプロジェクトの <i>xirr</i> は 10.985% であった。 PV は DF で検算して、100 となる。	<pre>pv 0 1 2 3, :_100 5 5 125 YEAR CF DF PV 0 _100 1 - 1 5 0.901022 4.50511 2 5 0.81184 4.0592 3 125 0.731486 91.4357 - - - 100</pre>

Worked Example

投資額:C=100

期間:T=20

配当: $A_1, A_2 \dots, A_n = 10$ (一定)

```
xirr (i.20),:_100,(19#10)
0.0744424
```

内部収益率 = 7.75%

Worked Example

NB. Example (Kaneko P69)

ある年末に 100 円の株を購入。

配当は各年末に 5 円。

丸 3 年後に 120 円で売却

$$\frac{5}{1+r} + \frac{5}{(1+r)^2} + \frac{125}{(1+r)^3} = 100$$

```
xirr 0 1 2 3,:_100 5 5 125
0.109851
```

	<i>Year</i>	<i>CF</i>
0	Y_0	-100
1	Y_1	5
2	Y_2	5
3	Y_3	125
	<i>IRR</i>	10.985%

Scripts .

```
df=: 3 : 0
```

```
NB. discount factor
```

```
NB. 1/(1+irr)^i
```

```
NB. Usage: (df/xirr) 0 1 2 3,:_100 5 5 125
```

```
%(>:xirr y)^i. # {y
```

```
)
```

```
pv_sub=: 3 : 0
```

```
NB. Usage: (pv_sub/df/xirr) 0 1 2 3,:_100 5 5 125
```

```
tmp=.({: y),. df y
```

```
tmp=. (i.# tmp),.tmp,. _,}. */"1 tmp
```

```
tmp, -, -, -, {:+/}.tmp
```

```
)
```

```
pv=: 3 : 0
```

```
NB. Usage: (pv/pv_sub/df/xirr) 0 1 2 3, :_100 5 5 125
```

```
tmp2=. pv_sub y
```

```
'YEAR CF DF PV',":tmp2
```

```
)
```

2.3 現在価値と将来価値

- プレゼントバリュー（現在価値）将来の財のキャッシュフローを現在の価値に割り戻した
もの
- ネットプレゼントバリュー（正味現在価値）将来の財のキャッシュフローから、投資額を差
し引いて、現在の価値に割り戻したもの

2.3.1 現在価値（資本回収係数）

$$R = \frac{r(1+r)^T}{(1+r)^T - 1} C$$

2.3.2 正味現在価値 NPV

NPV net Present Value

$$NPV = \sum_{i=1}^n \frac{CF}{(1+k)^i - 1}$$

NPV では要求収益率 (k) を明示する。

NVP = (投資の現在価値 - 投資コスト) = (投資の将来価値 × 割引係数 - 投資コスト)

- J の関数定義

Example

1000 万円を 5% で借りて 20 年で支払い。

loan 5 20
0.0802426

毎年の支払いは 80.24 万円

loan=: 3 : 0
NB. Usage: u 8(%) 20(years)
'R T'=: y %100 1
(R*(>:R)^T)%<:(>:R)^T
)

Example: 収益が一定の場合の現在価値 PV.

r=5%

T=20 年

A=10 A₁ = A₂ = ... = A_T = 10

C=100

vr : value-rate

vr 10 5 20
124.622

vr=: 3 : 0
'A R T'=. y % 1 100 1 NB. r 8--> 0.08
(A%R)*-. %(>:R)^T
)

$PV = C - V = 24.62\%$

$T = \infty \rightarrow V = \frac{A}{r}$

<p>npv Net present values</p>	<p>npv 1;12;0.10 5 0.09;24#1 9.74453</p> <p>パラメーター (4) imm = 0=advance, 1=arrears/ (貸付/借入) frq = payment frequency (e.g. 1=annual, 12=monthly) int = annual earned interest rate pay = payments per annum</p>	
<p>vt v to the power t</p>	<p>0.065 5 0.05 vt 5 6 7 8 0.729881 0.695125 0.662023 0.630499</p> <p>NB. t = durations NB. int = interest rate NB. NB. e.g. 0.065 5 0.05 vt 5 6 7 8 NB. = discount at durations 5 6 7 8 NB. interest 6.5% for 5 years, 5% thereafter</p>	<p>x vt y x is r(ex.0.05) (6.5% 5 年後は 5%) y is durations</p>

- Calc

100 億の投資計画

要求利益率 8%

CF 1 年目 (0), 2 年目 (5 億), 3 年目 (10 億)

プロジェクトの売却見込み: 3 年末 120 億

```
npv 0;1;0.08;_100 0 5 130
7.48489
```

```
npv 1;1;0.08;_100 0 5 130
6.93045
```

他の npv 計算のための関数 .

<p>NB. Example (Kaneko P69) ある年末に 100 円の株を購入。 配当は各年末に 5 円。 丸 3 年後に 120 円で売却 (同じ例題) + 要求収益率 8.0%</p> <ul style="list-style-type: none"> • 割引は xirr でなく要求利子率 8.0% を用いる • 現価合計 99.299 円 NPV=8.145 円 • 要求利益率より xirr が高いので NPV がプラスになる。 	<pre>8 npvx 0 1 2 3, : _100 5 5 125 year CF DF PV 0 _100 1 - 1 5 0.925926 4.62963 2 5 0.857339 4.28669 3 125 0.793832 99.229 - - - 108.145 npv 0;1;0.08 ; _100 5 5 125 8.14535 xirr 0 1 2 3, : _100 5 5 125 0.109851</pre>
---	---

100 億の投資計画

要求利益率 8%

CF 1 年目 (0), 2 年目 (5 億), 3 年目 (10 億)

プロジェクトの売却見込み: 3 年末 120 億

8 npvx 0 1 2 3, : _100 0 5 130

year	CF	DF	PV
------	----	----	----

0	_100	1	-
---	------	---	---

1	0	0.925926	0
---	---	----------	---

2	5	0.857339	4.28669
---	---	----------	---------

3	130	0.793832	103.198
---	-----	----------	---------

-	-	-	107.485
---	---	---	---------

現価合計 107.5 億

NPV 7.5 億の優良案件である

xirr 0 1 2 3, : _100 0 5 130

0.106663

xirr は 10.66%

npv 0;1;0.08;_100 0 5 130

7.48489

npv 1;1;0.08;_100 0 5 130

6.93045

2.3.3 将来価値 FV (再投資)

$1/(1 + \text{割引率})$ は df(discount factor) である。これを戻すと再投資レートに相当する。

```
% df 0 1 2 3, : _100 5 5 125
1 1.10985 1.23177 1.36708
```

irr を再投資レートとして複利運用して将来価値 (FV) をもとめる。

3年後の 137.7 が将来価値合計である

```
fv 0 1 2 3, : _100 5 5 125
YEAR CF ReInvest-Rate FV
1 _100 - -
2 5 1.23177 6.15885
3 5 1.10985 5.54926
4 125 1 125
- - - 136.708
```

2.4 割引配当モデル DDM

割引配当モデル DDM divided discount model

こちらは株式専門のようだ

2.4.1 シンプルなモデル

1. 要求利益率 11%
2. 配当 各期(年)5円
3. 売却予定価格 120円
4. この条件での現時点における妥当な株価を算出する。

取り敢えず *npvx* を用いて計算する。

*5

```
11 npvx 0 1 2 3, : _100 5 5 125
```

*5 *npvx* の _100 は使われていないのでどのような数字を入れてもよい。 *xirr* と入力フォーマットを合わせた

```

0  _100      1      -
1   5 0.900901  4.5045
2   5 0.811622  4.05811
3  125 0.731191 91.3989
-   -      - 99.9615

```

適正株価は 100 円

```

xirr 0 1 2 3, :_100 5 5 125
0.109851

```

```

npv0 0 1 2 3, :_100 5 5 125
0  _100      1      -
1   5 0.901022  4.50511
2   5 0.81184  4.0592
3  125 0.731486 91.4357
-   -      - 100

```

要求金利の代わりに *xirr* を用いた *npv0* の結果。(金利差がほとんど無いので微差)

2.4.2 多段階配当割引モデル

Example issue:kaneko P81

- 配当が変化する。10 円、2 年間は 10%、その後は 5% ずつ伸びる
- 要求収益率 12%
- 配当割引モデルによる理論株価を求める
- *xirr* は変則な *t* に対応できる。

```

+ / 10* (1.1^/1 2),1.05^/i.10
148.879

```

```

lrr  _100 ,10* (1.1^/1 2),(1.1^2)* 1.05^/>:i.10
|ill-formed number

```

```

xirr (i.13), : _100, 10* (1.1^/1 2),(1.1^2)* 1.05^/>:i.10
0.0965172

```

終期を定めないと npvx は使いづらい

12 npvx (i.13),: _100,

10* (1.1^/1 2),(1.1^2)* 1.05^/;>:i.10 npvx は概ね 100 年で収束する

0	_100	1	-
1	11	0.892857	9.82143
2	12.1	0.797194	9.64605
3	12.705	0.71178	9.04317
4	13.3403	0.635518	8.47797
5	14.0073	0.567427	7.9481
6	14.7076	0.506631	7.45134
7	15.443	0.452349	6.98563
8	16.2152	0.403883	6.54903
9	17.0259	0.36061	6.13972
10	17.8772	0.321973	5.75598
11	18.7711	0.287476	5.39623
12	19.7096	0.256675	5.05897
-	-	-	88.2736

12 npvx (i.103),: _100,

10* (1.1^/1 2),(1.1^2)* 1.05^/;>:i.100

96	1187.35	1.88395e_5	0.0223691
97	1246.72	1.6821e_5	0.0209711
98	1309.06	1.50187e_5	0.0196604
99	1374.51	1.34096e_5	0.0184316
100	1443.23	1.19729e_5	0.0172796
101	1515.4	1.069e_5	0.0161997
102	1591.17	9.54469e_6	0.0151872
-	-	-	163.93

npvm 多段階割引モデル

多段階割引モデル npvm も計算しておこう

理論株価 164 円

要求利子率 12%

配当 初期値 10 以降の伸び率 10%(2 年) 5% (3 年目以降)

3 年目で将来の割引が出るので 4 年目まで計算すればよい

12 10 npvm 10 5;2 4
144.691 19.4675 164.158

3 年目からの将来展望が 144 円, 手前の 2 年間で合計 164 円となる
更に 7 年目以降は 3% と見込んだ場合

12 10 npvm 10 5 3 ; 2 4 5
85.2765 52.3881 137.665

理論価格は 7 年目以降が 85 円、6 年間で合計 137 円と見込まれる
多段階割引モデルは定率成長モデルの積み上げである。

$$P_n = \frac{D_1}{k - g}$$

k: 要求配当率
g: 配当の増加率
 D_1 : 当初の配当額

```
12 10 npvm_sub1 10 5 3 ;2 4 5
```

```

rate discount cummurate
1      11 9.82143 9.82143
2     12.1 9.64605 19.4675
-----
3    12.705 9.04317 28.5106
4    13.3403 8.47797 36.9886
5    14.0073 7.9481 44.9367
6    14.7076 7.45134 52.3881
-----
7    15.1489 6.85257 59.2406
8    15.6033 6.30192 65.5425

```

2.5 金融不安定仮説と *xirr*

経済学者ハイマン・ミンスキー (1919-1996) が見直されている。ベラルーシの移民の子としてシカゴに生まれ、シカゴ大学を経て、ハーバードでレオンテフとシュンペーターの指導の下で博士の学位を得ているケインジアンである。セントルイスのワシントン大学で長く教え、銀行の経営にも携わった。

金融の動勢と経済に与える影響に関する鋭い洞察は金融不安定仮説として知られているが数量モデルを作成することはなく、生前に中央銀行に影響を与えることはなかった。

金融市場は不安定さを内在し、加熱しすぎた金融をポンジー過程と呼び、この崩壊がバブルであるとする。ポンジーは人名でネズミ講の考案者としても知られる。中央銀行がバブルに油を注ぎ、スーパールーズでバブル資金を供給したり、物価安定を重視しすぎて金融を軽視し、既にしぼんだスフレに「100年に一度」とコメントしたり、かれこれの中央銀行の失態が目立つようだ。

ミンスキーに関しては稿を改めるが、*xirr* が不安定な金利にも対応できることを確認しておこう。

```
xirr 0 1 2 3, : _100 5 5 125
0.109851
```

```
xirr 0 1 2 3 4 , : _100 5 5 0 50
```

-0.130876

半値になったとすると毎年マイナス 13% の預金に相当する。

3 References

金子誠一「証券アナリストのための数学再入門」ときわ総合サービス（株） 2004

関 玄「利息計算の手引き」銀行研修社 1979

縄田和満「理工系のためのファイナンス理論」東洋経済新報社 2003

付録 A J のインストールとソースコードの入手

A.1 J の入手とインストール

<http://www.jsoftware.com>

から DL する。プラットフォームは WIN/32,64 MAC/INTEL,PPC LINUX/32,64 PocketPC/W-CE4

インストーラに従ってインストールする。

J6 はレジストリーを遮断しているので USB,CDROM にコピーして利用できる。

A.2 ADDON の入手

J6 では Run/Package Manager で可能。

ネットに繋がっていなかったり、古いバージョンでは

<http://www.jsoftware.com/jal/j602/addons/>.

wiki/frontpage/Libraly/JAL からバージョンを確認して DL する。

A.3 スクリプト

http://homepage3.nifty.com/asagaya_avenue

J APL/workshop から DL する

付録 B xirr

NB. xirr

NB. internal rate of return by newton-raphson iteration.

NB.

NB. form: t xirr cf

NB. t = optional initial trial value (decimal rate)

NB. cf = cash flow as 2 row table:

NB. time

NB. payment

NB.

NB. tolerance 1e_5, max iterations 15

NB. example:

NB. time=. 0 1.2 4.7 5

NB. pay=. _100 10 8 120 (=cf)

```

NB. [x=. 0 xirr time,:pay
NB. 0.0713808
NB. +/ pay * (1+x) ^ -time
NB. 1.45661e_12
xirr=: 3 : 0
0 xirr y
:
t=. %>:x
'time cf'=. |."1 y
tol=. 1e_5
max=. 15
if. 2>#cf do. 'cash flow must have at least two elements' return. end.
if. -. *./_1 1 e.*cf do. 'no sign change in cash flow' return. end.
df=. cf * time
mp=. +/ . *
while.
  r=. t - (cf mp t ^ time) % df mp t ^ time - 1
  tol < |t-r do.
    t=. r
    if. 0=max=. <:max do. 'iterations exceeded' return. end.
end.
<:%r
)

```

$$df = .cf * time$$

$$mp = . + / . *$$

$$\frac{cf mp t^{times}}{df mp t^{times-1}}$$

付録 C ロケール

多くのファイルをロードするとき、関数の名前が重複することがある。

このとき J は後からロードした関数で警告なしで上書きする。

危険があるときはロケールを用いる。(スクリプトファイルの初めの方に指定されている事が多い)

整理タンスの各引き出しと思えばよい。

```
coclass 'asagaya'
```

このような記述がされたスクリプトファイルでは

fn0.asagay_

のように後ろに所属を明示して用いる。

fn0.z_ などと z_ が付いているものはグローバルに用いることができる。